


# Example 7 – Cable impact

- A cable is fixed at one end and has a concentrated mass at the other end. The mass has an initial velocity. In its motion, the cable impacts on a second cable fixed at both extremities.
- Cables have no resistance to bending or compression (only to traction).



45

## Geometric data and materials:

See slide. The cable material is elastic (traction only) and has a density of 8000, a Young's modulus of  $2 \times 10^{11}$  and a Poisson's ratio of 0.3.

## Numerical Solutions

### WRAP06

The longer cable uses 20 elements of type FUN3, the shorter one 4 elements. Contact is treated by the pinball method, with hierarchic pinballs of level 2. The mesh generation file is:

```

OPTI ECHO 1;
OPTI DIME 3 ELEM SEG2;
p0 = 0 0 0;
p1 = 10 0 0;
n = 20;
tol = 0.001;
bar1 = (p0 d n p1) coul vert;
p2 = 5 -1 -1;
pc = 5 0 -1;
p3 = 5 1 -1;
n2 = n / 5;
bar2 = (p2 d n2 p3) coul jaun;
elim tol (bar2 et pc);
pm = (manu pois p1) coul turq;
pa = -0.5 0 -0.5;

pb = -0.5 0 0.5;
app1 = (manu tri3 p0 pb pa) coul rose;
pc = 4.5 -1 -1.5;
pd = 5.5 -1 -1.5;
app2 = (manu tri3 p2 pc pd) coul rose;
pe = 4.5 1 -1.5;
pf = 5.5 1 -1.5;
app3 = (manu tri3 p3 pe pf) coul rose;
mesh = bar1 et bar2 et pm et app1 et app2 et app3;
tass mesh;
opti sauv form 'wrap06.mesh';
sauv form mesh;
opti trac psc ftra 'wrap06.mesh.ps';
trac qual mesh;
    
```

## The input file is:

```

WRAP - 06
*-----
ECHO
conv win
CAST mesh
*-----Problem type
TRID NONL LAGR
*-----Dimensioning
DIME
PT3L 26 PT6L 9 FUN3 24 PMAT 1 COQ1 3 ZONE 3
TERM
*-----Geometry
GEOM FUN3 bar1 bar2 PMAT pm COQ1 app1 app2 app3 TERM
*-----Geometric Complements
COMP EBAI 0.001 LECT bar1 bar2 TERM
EBAI 0.5 LECT pm TERM ! only for graphical representation
EBAI 1.0 LECT app1 app2 app3 TERM
*-----Material data
MATE FUNE RO 8000. YOUN 2.0E11 NU 0.3 ERUP 1.0
ELAS 2.0E11
TRAC 1 2.0E11 1.
LECT bar1 bar2 TERM
MASS 100 LECT pm TERM
VM23 RO 8000 YOUN 2.0E11 NU 0. ELAS 2.0E11
TRAC 1 2.0E11 1.0
LECT app1 app2 app3 TERM

*-----Boundary conditions
LINK COUP BLOC 123 LECT p0 p2 p3 TERM
PINB BODY MLEV 2 LECT bar1
BODY MLEV 2 LECT bar2
*-----Initial conditions
INIT VITE 3 -100 LECT p1 TERM
*-----Outputs
ECRI VITE TFRE 100.E-3
PICH ALIC TEMP FREQ 5
POIN LECT p1 pc TERM
*-----Options
OPTI NOTE CSTA 0.25E0
nocr lect app1 app2 app3 term
PINS CNOR
*-----Transient calculation
CALC TINI 0 TFIN 0.900
*-----ANIMATION
PLAY
CAME 1 EYE 5.00000E+00 0.00000E+00 2.51174E+01
!
Q 1.00000E+00 0.00000E+00 0.00000E+00 0.00000E+00
VIEW 0.00000E+00 0.00000E+00 -1.00000E+00 0.00000E+00
RIGH 1.00000E+00 0.00000E+00 0.00000E+00
UP 0.00000E+00 1.00000E+00 0.00000E+00
FOV 2.48819E+01
    
```

```

CAME 2 EYE 2.01169E+01 -1.88353E+01 8.04171E+00
! 0 7.35450E-01 5.62509E-01 1.29873E-01 3.54725E-01
VIEW -5.90102E-01 7.35256E-01 -3.33434E-01
RIGH 7.14606E-01 6.67874E-01 2.08043E-01
UP -3.75657E-01 1.15508E-01 9.19533E-01
FOV 2.48819E+01

scen
geom
  navi free
  line heou
  poin sphp
  pinb pare
  colo pape
  lima on

titl tit1 'EUROPLEXUS (C) Animation'
tit2 'Cable Impact'
tit3 'Author: P. Casadei'
slr cam1 1 nfra 30
trac offs fich avi nocl nfto 1091 fps 25 kfre 10 comp -1 rend

titl tit2 'Geometry and Parent Pinballs'
slr cam1 1 nfra 30
trac offs fich avi cont nocl rend

slr cam1 1 nfra 1
trac offs fich avi cont nocl rend

!slr cam1 1 cam2 2 nfra 100 cent 5 0 0
slr cam1 1 cam2 2 nfra 100 cent 5 0 -0.5
trac offs fich avi cont nocl rend

titl tit2 'Transient with descendent pinballs'
slr cam1 1 nfra 30
trac offs fich avi cont nocl rend

scen geom navi free
line heou
poin sphp
pinb cdes

!VECT SCCO FIEL VITE SCAL USER PROG 5 PAS 5 70 TERM
!text vsca SFAC 0.01
lima on

freq 0 tfre 1.E-3

slr cam1 2 nfra 1


gotr loop 899 offs fich avi cont nocl rend

go
trac offs fich avi cont rend

ENDPLAY
=====POST-TREATMENT
SUIT
Post-treatment
ECHO
RESU ALIC TEMP GARD PSCR
SORT GRAP
AXTE 1000.0 'Time [ms]'
*-----Curve definitions
COUR 1 'dx_p1' DEPL COMP 1 NOEU LECT p1 TERM
COUR 2 'dy_p1' DEPL COMP 2 NOEU LECT p1 TERM
COUR 3 'dz_p1' DEPL COMP 3 NOEU LECT p1 TERM
COUR 4 'dx_pc' DEPL COMP 1 NOEU LECT pc TERM
COUR 5 'dy_pc' DEPL COMP 2 NOEU LECT pc TERM
COUR 6 'dz_pc' DEPL COMP 3 NOEU LECT pc TERM
*-----Plots
trac 1 2 3 axes 1.0 'DISPL. [M]'
trac 4 5 6 axes 1.0 'DISPL. [M]'
*-----Results qualification
QUAL DEPL COMP 1 LECT p1 TERM REFE -4.41946E+0 TOLE 5.E-3
DEPL COMP 2 LECT p1 TERM REFE -2.42214E-2 TOLE 5.E-3
DEPL COMP 3 LECT p1 TERM REFE -1.76791E+0 TOLE 5.E-3
DEPL COMP 1 LECT pc TERM REFE 0.00000E+0 TOLE 5.E-3
DEPL COMP 2 LECT pc TERM REFE 0.00000E+0 TOLE 5.E-3
DEPL COMP 3 LECT pc TERM REFE 0.00000E+0 TOLE 5.E-3
=====
FIN

```

The initial configuration (with parent pinballs shown) and the final configuration are:



An example of intermediate velocities:

